Isometrías

Práctica de AlgebraLineal, E.U.A.T., 2004/05

Una isometría es un tipo especial de aplicación lineal: es una aplicación lineal que "mueve las cosas rígidamente". Precisamente, una aplicación lineal f de un espacio vectorial V en V es una isometría cuando, para cualesquiera vectores u,v de V, se cumple que <f(u),f(v)>=<u,v> (trataremos siempre con el producto escalar usual). Como se ha visto en teoría, esto significa que una isometría no "deforma" las cosas; puede girarlas en torno a un punto o una recta, reflejarlas, o las dos cosas a la vez, pero no puede, por ejemplo, "aplastarlas".

En esta práctica vamos a construir algunas isometrías; es decir, vamos a dar la matriz que las representa, de forma que podamos dar la imagen por la isometría de cualquier vector que queramos. Lo haremos de una forma u otra dependiendo del tipo de isometría. Sólo trataremos las isometrías en dos y tres dimensiones (en R2 y en R3), que son las que podemos imaginar más fácilmente. Para entender esta práctica es necesario entender bien la práctica sobre aplicaciones lineales.

■ Matrices ortogonales

La matriz de una isometría respecto de *una base ortonormal* tiene una propiedad curiosa: su inversa es igual que su traspuesta (o lo que es equivalente, el producto de ella por su traspuesta da la matriz identidad). La matrices que cumplen esto se llaman **matrices ortogonales**. Cuando calculemos la matriz en la base usual de varias isometrías en los apartados siguientes,podréis comprobar que son verdaderamente matrices ortogonales.

■ Isometrías en el plano

■ Una rotación

Para hablar de rotaciones siempre tomaremos los ángulos en radianes: un ángulo de t radianes corresponde a t*360/(2*Pi) grados. Por ejemplo, Pi radianes son 180º (media vuelta), Pi/3 radianes son 60º...

La matriz de una rotación (siempre con respecto al origen, ya que aquí tratamos sólo con aplicaciones lineales, para las que f(0)=0) es muy simple. Si el ángulo de la rotación, en sentido contrario a las agujas del reloj, es t (*en radianes*), entonces su matriz es

```
rotacion={{Cos[t],−Sin[t]},{Sin[t],Cos[t]}};
MatrixForm[rotacion]
```

$$\begin{pmatrix} \mathrm{Cos}[t] & -\mathrm{Sin}[t] \\ \mathrm{Sin}[t] & \mathrm{Cos}[t] \end{pmatrix}$$

Desde luego, para hallar la matriz de una rotación en el sentido de las agujas del reloj no hay más que tomar t negativo.

Por ejemplo, la matriz de una rotación de ángulo Pi/3 (en radianes) es:

```
t = Pi / 3;
rotacion // MatrixForm
```

$$\begin{pmatrix} \frac{1}{2} & -\frac{\sqrt{3}}{2} \\ \frac{\sqrt{3}}{2} & \frac{1}{2} \end{pmatrix}$$

Desde luego, usando la matriz podemos calcular la imagen de cualquier punto:

```
rotacion.{2, 5}
```

$$\left\{1-\frac{5\sqrt{3}}{2},\ \frac{5}{2}+\sqrt{3}\right\}$$

Comprobamos que la matriz de la rotación es ortogonal (si cumple $A.A^T = I$):

```
rotacion.Transpose[rotacion] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

■ Una simetría

Queremos encontrar la matriz de una simetría S con respecto a una recta. Por ejemplo, tomemos la recta 3x−y=0.Un vector director (un vector a lo largo de la recta) es por ejemplo (1,3), y como lo que hace una simetría el plano es "reflejar" sobre la recta de la que se trate, vemos que S(1,3) = (1,3) (porque la recta no se mueve). Por otra parte, un vector perpendicular a la recta es (3,−1)(cualquier vector perpendicular a (1,3) vale, claro). Como los vectores pependiculares a la recta se reflejan, sabemos que S(3,−1)=(−3,1)Con esto ya sabemos la matriz de S en la base {(1,3),(3,−1)}.Es

```
matriz1 = Transpose[{{1, 0}, {0, -1}}]
```

{{1, 0}, {0, −1}}

Para calcular ahora la matriz de la simetría con respecto a la base usual no hay más que multiplicar por las matrices de cambio de base adecuadas (recordad la práctica sobre aplicaciones lineales):

```
cambio = Transpose[{{1, 3}, {3, -1}}];
matriz2 = cambio.matriz1.Inverse[cambio];

MatrixForm[matriz2]
```

$$\begin{pmatrix} -\frac{4}{5} & \frac{3}{5} \\ \frac{3}{5} & \frac{4}{5} \end{pmatrix}$$

■ Isometrías en el espacio de tres dimensiones

■ Rotación de ángulo t respecto a una recta

Seguiremos la misma idea que antes: hallamos la matriz de la rotación respecto de una base particularmente fácil, y luego cambiamos a la base usual. Supongamos, por ejemplo, que queremos calcular la rotación de ángulo Pi/2 en torno a la recta generada por el vector u=(1,2,−1). Si R es la rotación que buscamos, sabemos que R(1,2,−1)=(1,2,−1)(porque al girar alrededor de la recta, la recta no cambia). Si encontramos dos vectores ortonormales y perpendiculares a la recta, sabemos que R actúa sobre estos vectores como una rotación en dos dimensiones (que hemos calculado antes). Busquemos estos dos vectores:

```
{v, w} = NullSpace[{{1, 2, -1}}]
```

```
{{1, 0, 1}, {-2, 1, 0}}
```

Esto nos da una base del subespacio de vectores perpendiculares a la recta v=(1,0,1) y w=(−2,1,0).Pero necesitamos además que estos vectores sean ortonormales (para que la matriz de la rotación sea la que queremos): para eso podemos, por ejemplo, aplicar Gramm−Schmith

```
w = w - (w.v) / (v.v) v
```

```
{-1, 1, 1}
```

Así que cambiamos w=(−2,1,0)por w=(−1,1,1).Entonces normalizamos los vectores v y w (los dividimos por su norma para que tengan norma 1) y tenemos la base deseada:

```
u = {1, 2, -1}
v = v / Sqrt[v.v]
w = w / Sqrt[w.w]
```

```
{1, 2, -1}
```

$$\left\{\frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}\right\}$$

$$\left\{-\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right\}$$

La matriz en esta base es (ya que sabemos que R(u)=u, y que R actúa sobre v, w como una rotación en R2):

```
t = Pi / 2;
rotacion1 = {{1, 0, 0}, {0, Cos[t], -Sin[t]}, {0, Sin[t], Cos[t]}};
MatrixForm[rotacion1]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 0 & -1 \\ 0 & 1 & 0 \end{pmatrix}$$

Ahora podemos, como antes, calcular la matriz de R en la base usual:

```
(* Matriz de cambio de la base {u,v,w} a la base usual *)
cambio = Transpose[{u, v, w}];
(* La matriz de R en la base usual *)
rotacion2 = cambio.rotacion1.Inverse[cambio]
MatrixForm[rotacion2]
```

$$\left\{\left\{\frac{1}{6}, \frac{1}{3}-\frac{1}{\sqrt{6}}, -\frac{1}{6}-\sqrt{\frac{2}{3}}\right\}, \left\{\frac{1}{3}+\frac{1}{\sqrt{6}}, \frac{2}{3}, -\frac{1}{3}+\frac{1}{\sqrt{6}}\right\}, \left\{-\frac{1}{6}+\sqrt{\frac{2}{3}}, -\frac{1}{3}-\frac{1}{\sqrt{6}}, \frac{1}{6}\right\}\right\}$$

$$\begin{pmatrix} \frac{1}{6} & \frac{1}{3}-\frac{1}{\sqrt{6}} & -\frac{1}{6}-\sqrt{\frac{2}{3}} \\ \frac{1}{3}+\frac{1}{\sqrt{6}} & \frac{2}{3} & -\frac{1}{3}+\frac{1}{\sqrt{6}} \\ -\frac{1}{6}+\sqrt{\frac{2}{3}} & -\frac{1}{3}-\frac{1}{\sqrt{6}} & \frac{1}{6} \end{pmatrix}$$

Podéis comprobar que su matriz es ortogonal :

```
rotacion2.Transpose[rotacion2] // Simplify // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

■ Simetría respecto a un plano

Para calcular la simetría con respecto a un plano (en la base canónica) podemos hacer algo parecido: elegimos una base que tenga un vector perpendicular al plano y dos vectores del plano, y entonces es muy fácil calcular su matriz en esta base; luego pasamos a la base canónica.

Calculemos, por ejemplo, la simetría respecto al plano dado por x+y+2z=0. Un vector perpendicular es (1,1,2) (se ve a ojo, son los coeficientes de la ecuación del plano).

```
u = {1, 1, 2};
```

Conseguir una base del plano es fácil:

```
{v, w} = NullSpace[{u}];
v
w
```

{-2, 0, 1}

{-1, 1, 0}

Escogemos pues la base {u,v,w}. La matriz en esta base es (sabiendo que S(u)= −u, S(v)=v, S(w)=w):

```
simetria1 = {{-1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
MatrixForm[simetria1]
```

$$\begin{pmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

La matriz en la base usual se obtiene como antes:

```
cambio = Transpose[{u, v, w}];
simetria2 = cambio.simetria1.Inverse[cambio];
MatrixForm[simetria2]
```

$$\begin{pmatrix} \frac{2}{3} & -\frac{1}{3} & -\frac{2}{3} \\ -\frac{1}{3} & \frac{2}{3} & -\frac{2}{3} \\ -\frac{2}{3} & -\frac{2}{3} & -\frac{1}{3} \end{pmatrix}$$

Podemos comprobar que simetria2 hace lo que debe con los vectores que escogimos antes como base:

```
simetria2.u
```

```
{-1, -1, -2}
```

```
simetria2.v
```

```
{-2, 0, 1}
```

```
simetria2.w
```

```
{-1, 1, 0}
```

■ Ejercicios

Todas las matrices que se pida calcular a continuación se refieren a la matriz en la base canónica del espacio correspondiente.

1. ¿Es la siguiente matriz ortogonal?
{{1,2,3},{0,2,1},{2,1,−7}}

2. ¿Cuál es la matriz de la simetría con respecto al eje vertical en R2?

3. ¿Cuál es la matriz de la rotación en R2 de media vuelta en el sentido de las agujas del reloj?

4. Calcula la matriz de la rotación en tres dimensiones de ángulo Pi/4 alrededor de la recta de ecuaciones {x−y=0,x+y+2z=0}.

5. Calcula la matriz de la simetría respecto al plano x=2y en R3.